Init

In[1]:=
```
$TextStyle = {FontFamily → "Helvetica", FontSize → 13};
```

In[2]:=
```
plotopt = Sequence @@ {GridLines → Automatic, Frame → True,
    FrameStyle → Thickness[0.0025], BaseStyle → {FontSize → 12}};
```

In[3]:=
```
plotoptn[n_Integer ?(# > 0 ∧ # < 6 &)] := Sequence @@
   {GridLines → Automatic, Frame → True, FrameStyle → Thickness[0.0025], PlotStyle →
     Take[{Purple, Brown, Darker[Green], Blue, Red}, -n], BaseStyle → {FontSize → 12}};
plotoptn[n_Integer ?(# ≤ 0 ∨ # ≥ 6 &)] := plotopt
```

In[5]:=
```
mylegend[labels_List, pos_ : Right] :=
 {Placed[LineLegend[labels, LabelStyle → {FontSize → 11},
    LegendMargins → 2, LegendFunction → (Framed[#, Background → White] &)], pos]}
```

Coupled Cavity Electric Fields

In[6]:=



Out[6]=



In[7]:=
```
(*
E0 = Toward input mirror;
E1 = Toward middle mirror;
E2 = Toward end mirror;
E3 = Back to middle mirror;
E4 = Back to input mirror;
E5 = Trans through input mirror;
E6 = Trans through end mirror;
l1 = length of first cavity;
l2 = length of second cavity;
k = wavenumber;
ϕ1 = the single pass tuning of the input mirror;
ϕ2 = the single pass tuning of the central mirror;
ϕ3 = the single pass tuning of the end mirror;
θ1 = k l1;
θ2 = k l2;
*)
```

In[8]:=
```
Clear[k]
```

$$M2 = \begin{pmatrix} 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ t1\, e^{-I\,\theta 1} & 0 & 0 & 0 & r1\, e^{-I\,\theta 1}\, e^{-I\,2\,\phi 1} & 0 & 0 \\ 0 & t2\, e^{-I\,\theta 2} & 0 & r2\, e^{-I\,\theta 2}\, e^{-I\,2\,\phi 2} & 0 & 0 & 0 \\ 0 & 0 & r3\, e^{-I\,\theta 2}\, e^{I\,2\,\phi 3} & 0 & 0 & 0 & 0 \\ 0 & -r2\, e^{-I\,\theta 1}\, e^{I\,2\,\phi 2} & 0 & t2\, e^{-I\,\theta 1} & 0 & 0 & 0 \\ -r1\, e^{I\,2\,\phi 1} & 0 & 0 & 0 & t1 & 0 & 0 \\ 0 & 0 & t3 & 0 & 0 & 0 & 0 \end{pmatrix};$$

```
(* Solve the systems of equations *)
G2 = Inverse[IdentityMatrix[7] - M2] /. {ϕ1 → 0, ϕ2 → 0, ϕ3 → 0};
(* All mirrors perfectly tuned *)
```

Electric field transfer functions

In[11]:=
```
(* Ecav to Erefl *)
E2toE5 = Simplify[Part[G2, 6, 3]]
```

Out[11]=
$$\frac{e^{i\,(\theta 1+\theta 2)}\, r3\, t1\, t2}{e^{2\,i\,(\theta 1+\theta 2)} + e^{2\,i\,\theta 2}\, r1\, r2 - e^{2\,i\,\theta 1}\, r2\, r3 - r1\, r3\, \left(r2^2 + t2^2\right)}$$

In[12]:=
```
(* Ein to Erefl *)
E0toE5 = Simplify[Part[G2, 6, 1]]
(* Ein to Eitm (incident on middle mirror)*)
E0toE1 = Simplify[Part[G2, 2, 1]]
(* Ein to Epop *)
E0toE4 = Simplify[Part[G2, 5, 1]]
(* Ein to Ecav *)
E0toE2 = Simplify[Part[G2, 3, 1]]
(* Ein to Etrans *)
E0toE6 = Simplify[Part[G2, 7, 1]]
```

Out[12]=
$$\frac{-e^{2i(\theta 1+\theta 2)}\,r1 + e^{2i\theta 1}\,r1\,r2\,r3 - e^{2i\theta 2}\,r2\left(r1^2+t1^2\right) + r3\left(r1^2+t1^2\right)\left(r2^2+t2^2\right)}{e^{2i(\theta 1+\theta 2)} + e^{2i\theta 2}\,r1\,r2 - e^{2i\theta 1}\,r2\,r3 - r1\,r3\left(r2^2+t2^2\right)}$$

Out[13]=
$$\frac{e^{i\theta 1}\left(e^{2i\theta 2} - r2\,r3\right)t1}{e^{2i(\theta 1+\theta 2)} + e^{2i\theta 2}\,r1\,r2 - e^{2i\theta 1}\,r2\,r3 - r1\,r3\left(r2^2+t2^2\right)}$$

Out[14]=
$$\frac{t1\left(-e^{2i\theta 2}\,r2 + r3\left(r2^2+t2^2\right)\right)}{e^{2i(\theta 1+\theta 2)} + e^{2i\theta 2}\,r1\,r2 - e^{2i\theta 1}\,r2\,r3 - r1\,r3\left(r2^2+t2^2\right)}$$

Out[15]=
$$\frac{e^{i(\theta 1+\theta 2)}\,t1\,t2}{e^{2i(\theta 1+\theta 2)} + e^{2i\theta 2}\,r1\,r2 - e^{2i\theta 1}\,r2\,r3 - r1\,r3\left(r2^2+t2^2\right)}$$

Out[16]=
$$\frac{e^{i(\theta 1+\theta 2)}\,t1\,t2\,t3}{e^{2i(\theta 1+\theta 2)} + e^{2i\theta 2}\,r1\,r2 - e^{2i\theta 1}\,r2\,r3 - r1\,r3\left(r2^2+t2^2\right)}$$

Advanced LIGO Parameters

In[17]:=
```
params = {f → 0, θ1 → 2 π f / c Ls, θ2 → 2 π f / c L,
    r1 → √(1 - T1 - A1), r2 → √(1 - T2 - A2), r3 → √(1 - T3 - A3),
    t1 → √T1, t2 → √T2, t3 → √T3,
    A1 → Aprm, A2 → Atm, A3 → Atm,
    T1 → Tprm};
values = {c → 3 * 10^8, Ls → 56, L → 4000,
    (*T1→0.031,*)T2 → 0.015, T3 → 5 * 10^-6};
```

Denominator

In[19]:=
```
denom = Simplify[(e^(2 I (θ1+θ2)) - e^(2 I θ2) r1 r2 - e^(2 I θ1) r2 r3 + r1 r3 (r2^2 + t2^2)) e^(-2 I (θ1+θ2)) //. params]
```

Out[19]=
$$1 - \sqrt{1 - \text{Atm} - \text{T2}}\ \sqrt{1 - \text{Atm} - \text{T3}} - \sqrt{1 - \text{Atm} - \text{T2}}\ \sqrt{1 - \text{Aprm} - \text{Tprm}} - (-1 + \text{Atm})\ \sqrt{1 - \text{Atm} - \text{T3}}\ \sqrt{1 - \text{Aprm} - \text{Tprm}}$$

In[20]:=
```
Abs[E0toE5 //. params /. values /. {Atm → 30 × 10^-6, Aprm → 0.005}]
```

Out[20]=
$$\text{Abs}\left[\frac{0.00745847 - 0.00756082\ \sqrt{0.995 - \text{Tprm}}}{0.00756082 + 0.992457\ \sqrt{0.995 - \text{Tprm}} - \frac{99\,997\ \sqrt{\frac{199\,993}{5}}\ \sqrt{0.995 - \text{Tprm}}}{20\,000\,000}}\right]$$

Plots

Plot parameters

In[56]:=
```
P0 = 39.2;(* W, input power incident on PRM *)
PRG0 = 57.2; (* power recycling gain *)
Parm0 = 261 000; (* W *)
refl0 = 0.09; (* carrier power reflection *)
Tprm0 = 0.031; (* PRM transmission *)
Aprm0 = 0.005;
(* PRC losses *)
```

Plot ranges

In[24]:=
```
Tprmlow = 0.01;
Tprmhigh = 0.05;
Atmlow = 10;(* ppm *)
Atmhigh = 50;(* ppm *)
dAtm = 10;(* ppm *)
```

Plot lines

In[29]:=
```
lineStyle = {Black, Dashed};
line1 = Line[{{Tprm0, 0}, {Tprm0, 1 000 000}}];
linerefl = Line[{{Tprmlow, refl0}, {Tprmhigh, refl0}}];
```

Find test mass loss solutions for measurements

Atm = Single-optic test mass loss

In[59]:=
```
FindRoot[
 refl0 == Abs[E0toE5 //. params /. values /. {Tprm → Tprm0, Aprm → Aprm0, Atm → Atm1 10^-6}]^2,
 {Atm1, 20}]
```

Out[59]= {Atm1 → 20.1364}

In[60]:=
```
FindRoot[
 PRG0 == Abs[E0toE4 //. params /. values /. {Tprm → Tprm0, Aprm → Aprm0, Atm → Atm1 10^-6}]^2,
 {Atm1, 20}]
```

FindRoot : The line search decreased the step size to within tolerance specified by AccuracyGoal and PrecisionGoal but was unable to find a sufficient decrease in the merit function. You may need more than MachinePrecision digits of working precision to meet these tolerances.

Out[60]= {Atm1 → 16.7457}

In[61]:=
```
FindRoot[Parm0 ==
  P0/2 Abs[E0toE2 //. params /. values /. {Tprm → Tprm0, Aprm → Aprm0, Atm → Atm1 10^-6}]^2,
 {Atm1, 20}]
```

FindRoot : The line search decreased the step size to within tolerance specified by AccuracyGoal and PrecisionGoal but was unable to find a sufficient decrease in the merit function. You may need more than MachinePrecision digits of working precision to meet these tolerances.

Out[61]= {Atm1 → 22.8397}

In[55]:=
```
FindRoot[
 {
  Parm0 ==
   P0/2 Abs[E0toE2 //. params /. values /. {Tprm → Tprm0, Aprm → Aprm0, Atm → Atm1 10^-6}]^2,
  PRG0 == Abs[E0toE4 //. params /. values /. {Tprm → Tprm0, Atm → Atm1 10^-6}]^2
 },
 {{Atm1, 20, 0, 100}, {Aprm, 0.01, 0, 0.05}}
]
```

FindRoot : The line search decreased the step size to within tolerance specified by AccuracyGoal and PrecisionGoal but was unable to find a sufficient decrease in the merit function. You may need more than MachinePrecision digits of working precision to meet these tolerances.

Out[55]= {Atm1 → 23.0212, Aprm → 0.0017102}

Arm losses vs Power levels

In[45]:=
```
p1 = Plot[
  Abs[E0toE5 //. params /. values /. {Tprm → Tprm0, Aprm → Aprm0, Atm → Atm1 10^-6}]^2,
  {Atm1, Atmlow, Atmhigh},
  ImageSize → Large,
  AxesLabel → {"Single test mass loss Atm [ppm]"},
  PlotLabel → Style["P_refl/P_0", Bold],
  Epilog → {Directive[lineStyle], line1},
  GridLines → {{refl0}},
  GridLinesStyle → Dashed
 ]
p2 = Plot[
  Abs[E0toE4 //. params /. values /. {Tprm → Tprm0, Aprm → Aprm0, Atm → Atm1 10^-6}]^2,
  {Atm1, Atmlow, Atmhigh},
  ImageSize → Large,
  AxesLabel → {"Single test mass loss Atm [ppm]"},
  PlotLabel → Style["PRG", Bold],
  Epilog → {Directive[lineStyle], line1},
  GridLines → {{PRG0}},
  GridLinesStyle → Dashed
 ]
p2 = Plot[
  P0/2 Abs[E0toE2 //. params /. values /. {Tprm → Tprm0, Aprm → Aprm0, Atm → Atm1 10^-6}]^2,
  {Atm1, Atmlow, Atmhigh},
  ImageSize → Large,
  AxesLabel → {"Single test mass loss Atm [ppm]"},
  PlotLabel → Style["PRG", Bold],
  Epilog → {Directive[lineStyle], line1},
  GridLines → {{Parm0}},
  GridLinesStyle → Dashed
 ]
```

Out[45]=



Out[46]=



Out[47]=



PRM Transmission vs Power levels

In[48]:=
```
p1 = Plot[
  Evaluate[Table[
    Legended[Abs[E0toE5 //. params /. values /. {Atm → Atms 10^-6, Aprm → Aprm0}]^2, Atms],
    {Atms, Atmlow, Atmhigh, dAtm}]],
  {Tprm, Tprmlow, Tprmhigh},
  ImageSize → Large,
  AxesLabel → Automatic,
  PlotLabel → Style["P_refl/P_0", Bold],
  Epilog → {Directive[lineStyle], line1},
  GridLines → {{refl0}},
  GridLinesStyle → Dashed,
  PlotRange → {0, 0.2},
  PlotLegends → LineLegend[, LegendLabel → "Single test mass loss Atm [ppm]"]
 ]
p2 = Plot[
  Evaluate[Table[
    Legended[Abs[E0toE4 //. params /. values /. {Atm → Atms 10^-6, Aprm → Aprm0}]^2, Atms],
    {Atms, Atmlow, Atmhigh, dAtm}]],
  {Tprm, Tprmlow, Tprmhigh},
  ImageSize → Large,
  AxesLabel → Automatic,
  PlotLabel → Style["PRG", Bold],
  Epilog → {Directive[lineStyle], line1},
  GridLines → {{PRG0}},
  GridLinesStyle → Dashed,
  PlotLegends → LineLegend[, LegendLabel → "Single test mass loss Atm [ppm]"]
 ]
p3 = Plot[
  Evaluate[Table[
    Legended[P0/2 Abs[E0toE2 //. params /. values /. {Atm → Atms 10^-6, Aprm → Aprm0}]^2, Atms],
    {Atms, Atmlow, Atmhigh, dAtm}]],
  {Tprm, Tprmlow, Tprmhigh},
  ImageSize → Large,
  AxesLabel → Automatic,
  PlotLabel → Style["P_arm", Bold],
  Epilog → {Directive[lineStyle], line1},
```

```
  GridLines → {{Parm0}},
  GridLinesStyle → Dashed,
  PlotLegends → LineLegend[, LegendLabel → "Single test mass loss Atm [ppm]"]
 ]
(*p4=Plot[
  Evaluate[Table[
    Legended[Abs[E0toE6// .params/.values/.{Atm→Atms 10^-6,Aprm→Aprm0}]^2,Atms],
    {Atms,Atmlow,Atmhigh,dAtm}]],
  {Tprm,Tprmlow,Tprmhigh},
  ImageSize→Large,
  AxesLabel→Automatic,
  PlotLabel→Style["P_trans/P_0",Bold],
  Epilog→{Directive[lineStyle],line1},
  GridLines→Automatic,
  PlotLegends→LineLegend[,LegendLabel→"Single test mass loss Atm [ppm]"]
 ]*)
```

Out[48]=


P_refl /P_0
0.20
0.15
0.10
0.05
0.01
0.02
0.03
Single test mass loss Atm [ppm]
10
20
30
40
50

Out[49]=


PRG
80
70
60
50
40
30
0.02
0.03
0.04
0.05
Tprm
Single test mass loss Atm [ppm]
10
20
30
40
50

Out[50]=


P_arm
450 000
400 000
350 000
300 000
250 000
200 000
150 000
100 000
0.02
0.03
0.04
0.05
Tprm
Single test mass loss Atm [ppm]
10
20
30
40
50

In[41]:=
```
Manipulate[
 Column[
  {
   Style["E_refl/E_0", Bold],
   Plot[Abs[E0toE5 //. params /. values]^2, {T1, 0.01, 0.05}, ImageSize → Medium],
   Style["E_prc/E_0", Bold],
   Plot[Abs[E0toE4 //. params /. values]^2, {T1, 0.01, 0.05}, ImageSize → Medium],
   Style["E_cav/E_0", Bold],
   Plot[Abs[E0toE2 //. params /. values]^2, {T1, 0.01, 0.05}, ImageSize → Medium],
   Style["E_trans/E_0", Bold],
   Plot[Abs[E0toE6 //. params /. values]^2, {T1, 0.01, 0.05}, ImageSize → Medium]
  }
 ],
 (*Controls*)
 {Atm, 0, 100 × 10^-6},
 {Aprm, 0, 0.02}
]
```



Out[41]=

E_cav/E_0



E_trans/E_0